

G13BCF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G13BCF calculates cross correlations between two time series.

2 Specification

```
SUBROUTINE G13BCF(X, Y, NXY, NL, S, RO, R, STAT, IFAIL)
  INTEGER          NXY, NL, IFAIL
  real           X(NXY), Y(NXY), S, RO, R(NL), STAT
```

3 Description

Given two series x_1, x_2, \dots, x_n and y_1, y_2, \dots, y_n the routine calculates the cross correlations between x_t and lagged values of y_t :

$$r_{xy}(l) = \frac{\sum_{t=1}^{n-l} (x_t - \bar{x})(y_{t+l} - \bar{y})}{n s_x s_y}, \quad l = 0, 1, \dots, L$$

where

$$\bar{x} = \frac{\sum_{t=1}^n x_t}{n}$$

$$s_x^2 = \frac{\sum_{t=1}^n (x_t - \bar{x})^2}{n}$$

and similarly for y .

The ratio of standard deviations s_y/s_x is also returned, and a portmanteau statistic is calculated:

$$\text{STAT} = n \sum_{i=1}^L r_{xy}(l)^2.$$

Provided n is large, L much less than n , and both x_t, y_t are samples of series whose true autocorrelation functions are zero, then, under the null hypothesis that the true cross correlations between the series are zero, STAT has a χ^2 distribution with L degrees of freedom. Values of STAT in the upper tail of this distribution provide evidence against the null hypothesis.

4 References

- [1] Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* Holden-Day (Revised Edition)

5 Parameters

- 1: X(NXY) — *real* array *Input*
On entry: the n values of the x series.
- 2: Y(NXY) — *real* array *Input*
On entry: the n values of the y series.
- 3: NXY — INTEGER *Input*
On entry: the length of the time series, n .
Constraint: NXY \geq 2.

- 4:** NL — INTEGER *Input*
On entry: the maximum lag for calculating cross correlations, L .
Constraint: $1 \leq \text{NL} < \text{NXY}$.
- 5:** S — *real* *Output*
On exit: the ratio of the standard deviation of the y series to the standard deviation of the x series, s_y/s_x .
- 6:** R0 — *real* *Output*
On exit: the cross correlation between the x and y series at lag zero.
- 7:** R(NL) — *real* array *Output*
On exit: the cross correlations between the x and y series at lags 1 to L , $r_{xy}(l)$ for $l = 1, 2, \dots, L$.
- 8:** STAT — *real* *Output*
On exit: the statistic for testing for absence of cross correlation.
- 9:** IFAIL — INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1 . For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, $\text{NXY} \leq 1$,
 or $\text{NL} < 1$,
 or $\text{NL} \geq \text{NXY}$.

IFAIL = 2

One or both of the x and y series have zero variance and hence cross correlations cannot be calculated.

7 Accuracy

All computations are believed to be stable.

8 Further Comments

The time taken by the routine is approximately proportional to nL .

9 Example

The example program reads two time series of length 20. It calculates and prints the cross correlations up to lag 15 for the first series leading the second series and then for the second series leading the first series.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G13BCF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
INTEGER          NXYMAX, NLMAX
PARAMETER       (NXYMAX=20,NLMAX=15)
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
real           ROXY, ROYX, STATXY, STATYX, SXY, SYX
INTEGER          I, IFAIL, NL, NXY
*      .. Local Arrays ..
real           RXY(NLMAX), RYX(NLMAX), X(NXYMAX), Y(NXYMAX)
*      .. External Subroutines ..
EXTERNAL        G13BCF
*      .. Executable Statements ..
WRITE (NOUT,*) 'G13BCF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
*      Read series length and number of lags
READ (NIN,*) NXY, NL
IF (NXY.GT.2 .AND. NXY.LE.NXYMAX .AND. NL.GT.0 .AND. NL.LE.NLMAX)
+   THEN
*      Read series
READ (NIN,*) (X(I),I=1,NXY)
READ (NIN,*) (Y(I),I=1,NXY)
*      Call routine to calculate cross correlations between X and Y
IFAIL = 0
*
CALL G13BCF(X,Y,NXY,NL,SXY,ROXY,RXY,STATXY,IFAIL)
*
IFAIL = 0
*
*      Call routine to calculate cross correlations between Y and X
CALL G13BCF(Y,X,NXY,NL,SYX,ROYX,RYX,STATYX,IFAIL)
*
WRITE (NOUT,*)
WRITE (NOUT,*)
+   '          Between          Between'
WRITE (NOUT,*)
+   '          X and Y          Y and X'
WRITE (NOUT,*)
WRITE (NOUT,99999) 'Standard deviation ratio', SXY, SYX
WRITE (NOUT,*)
WRITE (NOUT,*) 'Cross correlation at lag'
WRITE (NOUT,99999) '          0', ROXY, ROYX
WRITE (NOUT,99998) (I,RXY(I),RYX(I),I=1,NL)
WRITE (NOUT,*)
WRITE (NOUT,99997) 'Test statistic          ', STATXY, STATYX
END IF
STOP
*
99999 FORMAT (1X,A,F10.4,F15.4)
99998 FORMAT (21X,I4,F10.4,F15.4)
99997 FORMAT (1X,A,F10.4,F15.4)

```

END

9.2 Program Data

G13BCF Example Program Data

20	15									
0.02	0.05	0.08	0.03	-0.05	0.11	-0.01	-0.08	-0.08	-0.11	
-0.18	-0.19	-0.09	0.03	0.10	0.15	-0.14	0.07	0.09	0.16	
3.18	3.21	3.26	3.25	3.08	3.01	3.06	3.17	3.12	3.04	
3.26	3.45	3.33	3.70	3.31	3.81	3.33	2.96	3.28	3.10	

9.3 Program Results

G13BCF Example Program Results

	Between X and Y	Between Y and X
Standard deviation ratio	2.0053	0.4987
Cross correlation at lag		
0	0.0568	0.0568
1	0.0438	-0.0151
2	-0.3762	0.3955
3	-0.4864	0.3417
4	-0.6294	0.5486
5	-0.3871	0.2291
6	-0.1690	0.3190
7	-0.0678	0.1980
8	0.0962	0.0438
9	0.0788	-0.1428
10	0.2910	-0.1376
11	0.0950	-0.0387
12	0.0547	-0.0380
13	0.1855	-0.1551
14	0.0243	-0.1536
15	0.0034	-0.0696
Test statistic	22.1269	17.2917
